


Application Note		Ref No: an2k-037	Last Modify 120209
テーマ	MPC-1000/N816 パルス発生(タッチパネルで点データ編集とフラッシュ ROM 保存)		
使用機器	MPC-1000 または MPC-N816、ステップモータ、タッチパネル		

■機器構成



■概要

- MPC-1000 または MPC-N816 ワンボードでパルス発生を行います。
- 1軸テーブルを CW 方向にピッチ移動(絶対座標移動)して原点に戻ります。移動先の点データはタッチパネルの 10 キーで編集します。
- 点データは MPC のフラッシュ ROM に書き込むので、バッテリーバックアップの無い MPC-1000、MPC-N816 でも電源オフで消失しません。

■関連資料

[アプリケーションノート] MPC-1000 パルス発生例

<http://deparonline.jp/mpc2000/ref/headline/appendix/pdf/an2k-023.pdf>

[技術情報] MPC-816 から MPC-N816 への移行資料 > MPC-N816 J6 コネクタ 簡易パルス発生例

<http://deparonline.jp/mpc2000/ref/headline/appendix/pdf/MPC-816toN816.pdf>

■MPC サンプルプログラム

- サブルーチン *PGA_HOME で原点復帰をします。
- サブルーチン *PNT_EDIT で点データの編集と保存をします。フラッシュ ROM 書き込みは FSP コマンドです。書き込まれたデータはパワーオン時に自動的に読み込まれます。(コマンドで読む場合は FLP)
- 移動先の点データとして P(100)、P(102)、P(104)、P(106)、P(108) を使用しています。P(110)は速度(FEED)の値です。
- 点データ(X(n),Y(n),U(n),Z(n))は符号付 4byte 長、MBK(n)は基本形が符号無し 2byte 長です。サイズを整合させるため MBK(n)を MBK(n~Lng)とキャストしています。~Lng でキャストすると MBK(n)を下位、MBK(n+1)を上位とします。点データを 2 つ飛びに使っているのは MBK(n~Lng)に番号を合わせているためで、特に意味はありません。

```
MEWNET 38400 1          /* タッチパネル通信設定
TIME 1000
MBK(0)=0                /* タッチパネル page0
MBK(300)=0              /* メッセージクリア

PGA "D" 0                /* パルス出力方式 0=CW/CCW(default), 1=PULSE/DIR
PGA "A" 10000           /* 最高速(pps)。パラメータ変更後は演算時間を要する
WAIT SW(204)==1        /* PGA レディー待ち

DO
  DO
    MSG$=""
    S_MBK MSG$ 90 12     /* タッチパネル位置表示クリア
    MBK(300)=1
    WAIT SW(71000)|SW(71001) /* [MOVE] ボタン or [PNT EDIT] ボタン
    IF SW(71000) THEN    /* [MOVE]
      BREAK              /* EXIT DO~LOOP
    ELSE
      GOSUB *PNT_EDIT
    END_IF
  LOOP

  GOSUB *PGA_HOME       /* 原点復帰

  MBK(300)=2            /* "移動中"メッセージ
  PGA "F" X(110)        /* スピード min0~max10
  FOR PNT=100 TO 108 STEP 2
    MSG$="MOVE to P"+STR$(PNT)+" "
    S_MBK MSG$ 90 12     /* タッチパネル 移動先表示

    PGA "M" X(PNT)      /* 絶対座標移動
    WAIT SW(204)==1     /* PGA レディー待ち
  NEXT

  MSG$="MOVE to ORG"
  S_MBK MSG$ 90 12     /* タッチパネル 移動先表示

  PGA "M" 0
  WAIT SW(204)==1
  TIME 500

LOOP

*PGA_HOME
MBK(300)=3              /* "原点復帰中"メッセージ
IF SW(198)==1 THEN     /* もしも原点 LS がオンしていたら退避移動をする
  PGA "F" 5            /* speed
```

```

PGA "R" 1000 /* 相対座標移動
WAIT SW(204)==1
END_IF
PGA "G" -4000 /* 定速移動 CCW方向 4Kpps
WAIT SW(198)==1 /* 原点LSオン待ち
OFF PGA /* PGA パルス発生停止
TIME 50 /* ディレイ必要
PGA "H" 0 /* この位置を0とする
PGA "V" /* ファームウェアバージョン番号取得
PRINT "PGA version=" V_PGA /* ファームウェアバージョン番号表示 (V_PGA は予約変数)
PGA "C" /* 現在位置取得
PRINT "Current Positon" V_PGA /* 現在位置表示
RETURN

*PNT_EDIT
FOR PNT=100 TO 110 STEP 2 /* 現在の位置データを MBK() にセットする
  MBK(PNT~Lng)=X(PNT)
NEXT
MBK(0)=1 /* タッチパネル page1

WAIT SW(71002)|SW(71003) /* "Ok" ボタン or "Cancel" ボタン
IF SW(71002) THEN /* "Ok" ボタン
  FOR PNT=100 TO 110 STEP 2
    X(PNT)=MBK(PNT~Lng) /* 設定内容を位置データにセットする
  NEXT

  FSP /* P(100)~(299) フラッシュROMに書き込み
        /* パワーオン時に読み込まれる

END_IF

MBK(0)=0 /* タッチパネル page0
RETURN

```

■タッチパネルデザイン



・ベース画面 1

DT300 メッセージ部品
 No1 (空白)
 No2 ボタンを押してください
 No3 移動中
 No4 原点復帰中

R100 スイッチ部品 オルタネート
 R101 スイッチ部品 モーメンタリ

DT90 データ部品
 データ形式 ASCII 12文字

・ベース画面 2

DT100 ~ DT110 データ部品
 データ形式 DEC(2W)

R102 スイッチ部品 モーメンタリ
 R103 スイッチ部品 モーメンタリ

キーボード部品 No0

MPC プログラム対応例

DT300	→ MBK(300)=MessageNo	: タッチパネルに登録してあるメッセージを表示
DT100	→ MBK(100~Lng)=X(100)	: MBK(100), MBK(101) ~ X(100)の値をコピー
DT90	→ S_MBK MSG\$ 90 12	: 文字列 MSG\$を MBK(90)を先頭に 12文字コピー
R100	→ SW(71000)	: SW(71000)ビット入力
R101	→ SW(71001)	: SW(71001)ビット入力

■実行中のタッチパネル画面



・最初の画面（ベース画面 1）

MOVE ボタンを押すと原点復帰し、その後ピッチ移動をします。


PNT EDIT ボタンで点データ編集画面になります。



・原点復帰中



・移動中




・点データ編集画面（ベース画面 2）

現在の各点の値を表示します。

数字をタッチすると10キーが現れ値を変更することができます。

Ok ボタンでフラッシュ ROM に書き込みます。



--- End Of Doc ---